

x86-64
Estruturas programáticas

Bib: Computer Systems: A Programmer's Perspective
x86-64 Machine-Level Programming (adenda ao cap. 3 do livro anterior)

Programação em Sistemas Computacionais

João Pedro Patriarca (joao.patriarca@isel.pt), Gabinete F.0.23 do edifício F

ISEL, ADEETC, LEIC

Agenda

- Estruturas programáticas em *assembly*:
  - Instruções de decisão: *if*, *if-then-else* , *?:* (operador ternário)
  - Instruções de ciclo: *do-while*, *while*, *for*
  - Instrução de múltipla decisão: *switch-case*

Instruções de decisão

IF-THEN e IF-THEN-ELSE

```
int if_then(int v) {
    if (v < 0) v = -v;
    return v;
}


int if_then_else(int v) {
    if (v >= 10) v += 1;
    else v = -v;
    return v;
}
```





```
if_then:
    mov     %edi, %eax
    test    %eax, %eax
    jns     .L0_if_then
    neg     %eax
.L0_if_then:
    ret

if_then_else:
    mov     %edi, %eax
    cmp     $10, %eax
    jl      .L0_if_then_else
    add     $1, %eax
    jmp     .L1_if_then_else
.L0_if_then_else:
    neg     %eax
.L1_if_then_else:
    ret
```

- Por norma, o salto condicional é implementado com a relação inversa

Instruções de decisão

Operador ternário

```
int ternary_op(int v) {
    return v >= -1 && v <= 1 ?
               0 :
               v*4;
}
```

```
ternary_op:
    lea     (,%edi,4), %eax
    cmp     $-1, %edi
    jl      .L0_ternary_op
    cmp     $1, %edi
    jg      .L0_ternary_op
    xor     %eax, %eax
.L0_ternary_op:
    ret
```

- Transforma a instrução de dois casos para um caso, produzindo inicialmente o resultado da cláusula *else* e alterando-o apenas se o resultado da comparação for verdadeiro

Instruções de ciclo
WHILE

```
int _while (uint v,
            uint idx,
            uint len) {
    ...
    v >>= idx;
    int cnt = 0;
    while (len != 0) {
        cnt += v & 1;
        len -= 1;
        v >>= 1;
    }
    return cnt;
}
```

```
_while:
    ...
    mov     %sil, %cl
    shr     %cl, %edi
    xor     %eax, %eax
    jmp     .L0_while
.L1_while:
    mov     %edi, %ecx
    and     $1, %ecx
    add     %ecx, %eax
    dec     %dx
    shr     $1, %edi
.L0_while:
    test    %dx, %dx
    jnz     .L1_while
    ret
```

- Para reduzir o número de saltos por iteração, a expressão de controlo é movida para o final do *while*

Instruções de ciclo

DO-WHILE

```
int _do_while(ulong v) {
    int cnt = 0, cnt_tmp;
    do {
        cnt_tmp = 0;
        while (v & 1) {
            cnt_tmp += 1;
            v >>= 1;
        }
        if (cnt_tmp > cnt)
            cnt = cnt_tmp;
        while (v != 0 &&
               (v & 1) == 0)
            v >>= 1;
    } while (v != 0);
    return cnt;
}
```



```
_do_while:
    xor     %eax, %eax # cnt = 0
.L5_do_while:
    xor     %edx, %edx # cnt_tmp = 0
    jmp     .L0_do_while
.L1_do_while:
    inc     %edx
    shr     $1, %rdi
.L0_do_while:
    test    $1, %rdi
    jnz     .L1_do_while
    cmp     %edx, %eax
    jge     .L2_do_while
    mov     %edx, %eax
    jmp     .L2_do_while
.L3_do_while:
    shr     $1, %rdi
.L2_do_while:
    test    %rdi, %rdi
    jz      .L4_do_while
    test    $1, %rdi
    jz      .L3_do_while
.L4_do_while:
    test    %rdi, %rdi
    jnz     .L5_do_while
    ret
```

Instruções de ciclo
FOR

```
int _for(int v[],
         int vsize) {
    int sum = 0;
    for (int i = 0;
             i < vsize;
             i++)
        sum += v[i];
    return sum;
}
```



```
_for:
    xor     %eax, %eax # sum = 0
    xor     %rdx, %rdx # i = 0
    jmp     .L0_for
.L1_for:
    add     (%rdi, %rdx, 4), %eax
    inc     %rdx
.L0_for:
    cmp     %esi, %edx
    jl      .L1_for
    ret
```

Instrução de múltipla decisão

SWITCH-CASE com tabela de CASES

```c
typedef enum {
    add, sub, mul, div, mod
} Operation;

int _switch_case(
    Operation op, int a, int b)
{
    int r;
    switch(op) {
        case add: r = a+b; break;
        case sub: r = a-b; break;
        case mul: r = a*b; break;
        case div: r = a/b; break;
        case mod: r = a%b; break;
        default: r = 0;
    }
    return r;
}
```

```
_switch_case:
    cmp     $4, %edi
    jg      .switch_end
    movabs  $table_cases, %rcx
    jmp     *(%rcx, %rdi, 8)
.case_add:  ...
    jmp     .switch_end
.case_sub:  ...
    jmp     .switch_end
.case_mul:  ...
    jmp     .switch_end
.case_div:  ...
    jmp     .switch_end
.case_mod:  ...
.switch_end:
    ret

    .section .rodata
table_cases:
    .quad    .case_add, .case_sub,
             .case_mul, .case_div,
             .case_mod
```

Instrução de múltipla decisão

SWITCH-CASE com tabela de JMPs

```
int _switch_case(
    Operation op,
    int a,
    int b)
{
    int r;
    switch(op) {
        case add: r = a+b; break;
        case sub: r = a-b; break;
        case mul: r = a*b; break;
        case div: r = a/b; break;
        case mod: r = a%b; break;
        default: r = 0;
    }
    return r;
}
```

```
_switch_case:
    cmp     $4, %edi
    jg      .switch_end
    movabs  $.table_jmp, %rcx
    lea     (%rcx, %rdi, 2), %rcx
    jmp     *%rcx
.table_jmp:
    jmp     .case_add
    jmp     .case_sub
    jmp     .case_mul
    jmp     .case_div
    jmp     .case_mod
.case_add: ...
    jmp     .switch_end
.case_sub: ...
    jmp     .switch_end
.case_mul: ...
    jmp     .switch_end
.case_div: ...
    jmp     .switch_end
.case_mod: ...
.switch_end:
    ret
```

Exercícios

1. Implemente a função `char * strcat(char * dst, const char * src)` que concatena a *string* `src` na *string* `dst`. A função retorna `dst`.
2. Implemente a função `int get_greater(int *a, int a_size)` que retorna o maior valor inteiro de uma sequência de valores inteiros com sinal
3. Implemente a função `uint get_ugreater(uint *a, int a_size)` que retorna o maior valor inteiro de uma sequência de valores inteiros sem sinal
4. Implemente a função `int index_of_fastest(PCar * car, int size)` que a partir de um *array* de carros retorna o índice do carro com maior velocidade. O tipo `Car` é caracterizado por:
   - Marca (*string*)
   - Modelo (*string*)
   - Fonte de energia (enumerado com os valores Gasolina, Gasóleo, Híbrido, Elétrico)
   - Velocidade máxima (inteiro)